import std.stdio;
import std.string;
import std.format;
import std.conv;
import std.typecons;
import std.algorithm;
import std.functional;
import std.bigint;
import std.numeric;
import std.array;
import std.math;
import std.range;
import std.container;
import std.concurrency;
import std.traits;
import std.uni;
import std.regex;
import core.bitop : popcnt;

alias Generator = std.concurrency.Generator;

enum long INF = long.max / 5;

struct P {
  long x, y;
}

long f(P[] rs, P[] us) {
  rs = rs.dup;
  us = us.dup;
  foreach (ref r; rs) {
    r.y += r.x;
  }
  foreach (ref u; us) {
    u.y += u.x;
  }
  us.sort!"a.y != b.y ? a.y < b.y : a.x < b.x";
  long res = INF;
  foreach (r; rs) {
    auto t = us.assumeSorted!"a.y != b.y ? a.y < b.y : a.x < b.x".upperBound(
        P(r.x - 1, r.y)
    );
    if (t.empty)
      continue;
    P p = t.front;
    if (p.y != r.y)
      continue;
    assert(p.x >= r.x);
    res.ch!min((p.x - r.x) * 10);
  }
  return res;
}

long g(P[] rs, P[] ls) {
  rs = rs.dup;
  ls = ls.dup;
  rs.sort!"a.y != b.y ? a.y < b.y : a.x < b.x";
  ls.sort!"a.y != b.y ? a.y < b.y : a.x < b.x";
  long i = 0;
  long j = 0;
  long res = INF;
  while (i < rs.length) {
    P r = rs[i++];
    while (j < ls.length && (ls[j].y < r.y || (ls[j].y == r.y && ls[j].x < r.x))) {
      j++;
    }
    if (j >= ls.length || ls[j].y > r.y) {
      continue;
    }
    P l = ls[j];
    res.ch!min((l.x - r.x) * 5);
  }
  return res;
}

void main() {
  long N;
  scanln(N);
  P[] us, rs, ls, ds;
  {
    P[]*[char] aa;
    aa['U'] = &us;
    aa['R'] = &rs;
    aa['L'] = &ls;
    aa['D'] = &ds;
    foreach (i; 0 .. N) {
      P p;
      char u;
      scanln(p, u);
      *aa[u] ~= p;
    }
  }

  long ans = INF;
  ans.ch!min(g(
      rs.map!(p => P(p.x, p.y)).array,
      ls.map!(p => P(p.x, p.y)).array,
  ));
  ans.ch!min(g(
      us.map!(p => P(p.y, p.x)).array,
      ds.map!(p => P(p.y, p.x)).array,
  ));
  ans.ch!min(f(
      rs.map!(p => P(p.x, p.y)).array,
      us.map!(p => P(p.x, p.y)).array,
  ));
  ans.ch!min(f(
      rs.map!(p => P(p.x, -p.y)).array,
      ds.map!(p => P(p.x, -p.y)).array,
  ));
  ans.ch!min(f(
      ls.map!(p => P(-p.x, p.y)).array,
      us.map!(p => P(-p.x, p.y)).array,
  ));
  ans.ch!min(f(
      ls.map!(p => P(-p.x, -p.y)).array,
      ds.map!(p => P(-p.x, -p.y)).array,
  ));

  if (ans == INF) {
    writeln("SAFE");
  } else {
    ans.writeln;
  }
}

// ----------------------------------------------

void times(alias fun)(long n) {
  // n.iota.each!(i => fun());
  foreach (i; 0 .. n)
    fun();
}

auto rep(alias fun, T = typeof(fun()))(long n) {
  // return n.iota.map!(i => fun()).array;
  T[] res = new T[n];
  foreach (ref e; res)
    e = fun();
  return res;
}

T ceil(T)(T x, T y) if (isIntegral!T || is(T == BigInt)) {
  // `(x+y-1)/y` will only work for positive numbers ...
  T t = x / y;
  if (y > 0 && t * y < x)
    t++;
  if (y < 0 && t * y > x)
    t++;
  return t;
}

T floor(T)(T x, T y) if (isIntegral!T || is(T == BigInt)) {
  T t = x / y;
  if (y > 0 && t * y > x)
    t--;
  if (y < 0 && t * y < x)
    t--;
  return t;
}

ref T ch(alias fun, T, S...)(ref T lhs, S rhs) {
  return lhs = fun(lhs, rhs);
}

unittest {
  long x = 1000;
  x.ch!min(2000);
  assert(x == 1000);
  x.ch!min(3, 2, 1);
  assert(x == 1);
  x.ch!max(100).ch!min(1000); // clamp
  assert(x == 100);
  x.ch!max(0).ch!min(10); // clamp
  assert(x == 10);
}

mixin template Constructor() {
  import std.traits : FieldNameTuple;

  this(Args...)(Args args) {
    // static foreach(i, v; args) {
    foreach (i, v; args) {
      mixin("this." ~ FieldNameTuple!(typeof(this))[i]) = v;
    }
  }
}

template scanln(Args...) {
  enum sep = " ";
  enum n = () {
    long n = 0;
    foreach (Arg; Args) {
      static if (is(Arg == class) || is(Arg == struct) || is(Arg == union)) {
        n += Fields!Arg.length;
      } else {
        n++;
      }
    }
    return n;
  }();
  enum fmt = n.rep!(() => "%s").join(sep);
  enum argsString = () {
    string[] xs = [];
    foreach (i, Arg; Args) {
      static if (is(Arg == class) || is(Arg == struct) || is(
          Arg == union)) {
        foreach (T; FieldNameTuple!Arg) {
          xs ~= "&args[%d].%s".format(i, T);
        }
      } else {
        xs ~= "&args[%d]".format(i);
      }
    }
    return xs.join(", ");
  }();
  void scanln(auto ref Args args) {
    string line = readln.chomp;
    static if (__VERSION__ >= 2074) {
      mixin(
          "line.formattedRead!fmt(%s);".format(argsString)
      );
    } else {
      mixin(
          "line.formattedRead(fmt, %s);".format(argsString)
      );
    }
  }
}

// fold was added in D 2.071.0
static if (__VERSION__ < 2071) {
  template fold(fun...) if (fun.length >= 1) {
    auto fold(R, S...)(R r, S seed) {
      static if (S.length < 2) {
        return reduce!fun(seed, r);
      } else {
        return reduce!fun(tuple(seed), r);
      }
    }
  }
}

// popcnt with ulongs was added in D 2.071.0
static if (__VERSION__ < 2071) {
  ulong popcnt(ulong x) {
    x = (x & 0x5555555555555555L) + (
        x >> 1 & 0x5555555555555555L);
    x = (
        x & 0x3333333333333333L) + (x >> 2 & 0x3333333333333333L);
    x = (x & 0x0f0f0f0f0f0f0f0fL) + (
        x >> 4 & 0x0f0f0f0f0f0f0f0fL);
    x = (
        x & 0x00ff00ff00ff00ffL) + (
        x >> 8 & 0x00ff00ff00ff00ffL);
    x = (
        x & 0x0000ffff0000ffffL) + (
        x >> 16 & 0x0000ffff0000ffffL);
    x = (
        x & 0x00000000ffffffffL) + (
        x >> 32 & 0x00000000ffffffffL);
    return x;
  }
}


Measured memory_bytes: 24296